
F08 – Least Squares and Eigenvalue Problems (LAPACK)

F08FQF (CHEEVD/ZHEEVD) – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold
italicised terms and other implementation-dependent details.

1 Purpose

F08FQF (CHEEVD/ZHEEVD) computes all the eigenvalues, and optionally all the eigenvectors, of a
complex Hermitian matrix. If the eigenvectors are requested, then it uses a divide and conquer algorithm
to compute eigenvalues and eigenvectors. However, if only eigenvalues are required, then it uses the
Pal–Walker–Kahan variant of the QL or QR algorithm.

2 Specification

SUBROUTINE F08FQF(JOB, UPLO, N, A, LDA, W, WORK, LWORK, RWORK,
1 LRWORK, IWORK, LIWORK, INFO)
ENTRY cheevd(JOB, UPLO, N, A, LDA, W, WORK, LWORK, RWORK,
1 LRWORK, IWORK, LIWORK, INFO)
INTEGER N, LDA, LWORK, LRWORK, IWORK(∗), LIWORK, INFO
real W(∗), RWORK(*)
complex A(LDA,∗), WORK(∗)
CHARACTER∗1 JOB, UPLO

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine computes all the eigenvalues, and optionally all the eigenvectors, of a complex Hermitian
matrix A. In other words, it can compute the spectral factorization of A as

A = ZΛZH ,

where Λ is a real diagonal matrix whose diagonal elements are the eigenvalues λi, and Z is the (complex)
unitary matrix whose columns are the eigenvectors zi. Thus

Azi = λizi for i = 1, 2, . . . , n.

4 References

[1] Golub G H and Van Loan C F (1996) Matrix Computations Johns Hopkins University Press (3rd
Edition), Baltimore

5 Parameters

1: JOB — CHARACTER*1 Input

On entry: indicates whether eigenvectors are computed as follows:

if JOB = ’N’, then only eigenvalues are computed;
if JOB = ’V’, then eigenvalues and eigenvectors are computed.

Constraint: JOB = ’N’ or ’V’.

2: UPLO — CHARACTER*1 Input

On entry: indicates whether the upper or lower triangular part of A is stored as follows:

if UPLO = ’U’, then the upper triangular part of A is stored;
if UPLO = ’L’, then the lower triangular part of A is stored.

Constraint: UPLO = ’U’ or ’L’.

3: N — INTEGER Input

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

4: A(LDA, ∗) — complex array Input/Output

Note: the second dimension of the array A must be at least max(1,N).

On entry: the n by n Hermitian matrix A. If UPLO = ’U’, the upper triangular part of A must
be stored and the elements of the array below the diagonal are not referenced; if UPLO = ’L’, the
lower triangular part of A must be stored and the elements of the array above the diagonal are not
referenced.

On exit: if JOB = ’V’, then this is overwritten by the unitary matrix Z which contains the
eigenvectors of A.

5: LDA — INTEGER Input

On entry: the first dimension of the array A as declared in the (sub)program from which F08FQF
(CHEEVD/ZHEEVD) is called.

Constraint: LDA ≥ max(1,N).

6: W(∗) — real array Output

Note: the dimension of the array W must be at least max(1,N).

On exit: the eigenvalues of the matrix A in ascending order.

7: WORK(∗) — complex array Output

Note: the dimension of the array WORK must be at least LWORK.

On exit: if LWORK > 0, then the real part of WORK(1) contains the required minimal size of
LWORK.

8: LWORK — INTEGER Input

On entry: the dimension of the array WORK as declared in the (sub)program from which F08FQF
(CHEEVD/ZHEEVD) is called.

Constraints:

if N ≤ 1, then LWORK ≥ 1;
if JOB = ’N’ and N > 1, then LWORK ≥ N + 1;
if JOB = ’V’ and N > 1, then LWORK ≥ N2 + 2 × N.

9: RWORK(∗) — real array Input

Note: the dimension of the array RWORK must be at least LRWORK.

On exit: if LRWORK > 0, then RWORK(1) contains the required minimal size of LRWORK.

10: LRWORK — INTEGER Input

On entry: the dimension of the array LRWORK as declared in the (sub)program from which F08FQF
(CHEEVD/ZHEEVD) is called.

Constraints:

if N ≤ 1, then LWORK ≥ 1;
if JOB = ’N’ and N > 1, then LWORK ≥ N;
if JOB = ’V’ and N > 1, then LWORK ≥ 3 × N2 + (4+2k) × N + 1 where k is the smallest
integer which satisfies 2k ≥ N.

11: IWORK(∗) — INTEGER array Input

Note: the dimension of the array IWORK must be at least LIWORK.

On exit: if LIWORK > 0, then IWORK(1) contains the required minimal size of LIWORK.

12: LIWORK — INTEGER Input

On entry: the dimension of the array IWORK as declared in the (sub)program from which F08FQF
(CHEEVD/ZHEEVD) is called.

Constraints:

if N ≤ 1, then LIWORK ≥ 1,
if JOB = ’N’ and N > 1, then LIWORK ≥ 1,
if JOB = ’V’ and N > 1, then LIWORK ≥ 5 × N + 2.

13: INFO — INTEGER Output

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = −i, the ith parameter had an illegal value. An explanatory message is output, and
execution of the program is terminated.

INFO > 0

If INFO = i, then the algorithm failed to converge; i indicates the number of elements of an
intermediate tridiagonal form which did not converge to zero.

7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix A + E, where

‖E‖2 = O(ε)‖A‖2,

and ε is the machine precision.

8 Further Comments

The real analogue of this routine is F08FCF (SSYEVD/DSYEVD).

9 Example

To compute all the eigenvalues and eigenvectors of the Hermitian matrix A, where

A =




1.0 + 0.0i 2.0 + 1.0i 3.0 + 1.0i 4.0 + 1.0i
2.0− 1.0i 2.0 + 0.0i 3.0 + 2.0i 4.0 + 2.0i
3.0− 1.0i 3.0− 2.0i 3.0 + 0.0i 4.0 + 3.0i
4.0− 1.0i 4.0− 2.0i 4.0− 3.0i 4.0 + 0.0i


 .

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details.
Please read the Users’ Note for your implementation to check the interpretation of these terms. As explained in the Essential
Introduction to this manual, the results produced may not be identical for all implementations.

* F08FQF Example Program Text.
* Mark 19 Release. NAG Copyright 1999.
* .. Parameters ..

INTEGER NIN, NOUT
PARAMETER (NIN=5,NOUT=6)
INTEGER NMAX, LDA
PARAMETER (NMAX=8,LDA=NMAX)
INTEGER LWORK, LIWORK, LRWORK
PARAMETER (LWORK=NMAX*NMAX+2*NMAX,LIWORK=2+5*NMAX,

+ LRWORK=4*NMAX*NMAX)
* .. Local Scalars ..

INTEGER I, IFAIL, INFO, J, N
CHARACTER JOB, UPLO

* .. Local Arrays ..
complex A(LDA,NMAX), WORK(LWORK)
real RWORK(LRWORK), W(NMAX)
INTEGER IWORK(LIWORK)

* .. External Subroutines ..
EXTERNAL X04DAF, cheevd

* .. Executable Statements ..
WRITE (NOUT,*) ’F08FQF Example Program Results’

* Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN

READ (NIN,*) UPLO
*
* Read A from data file
*

IF (UPLO.EQ.’U’) THEN
READ (NIN,*) ((A(I,J),J=I,N),I=1,N)

ELSE IF (UPLO.EQ.’L’) THEN
READ (NIN,*) ((A(I,J),J=1,I),I=1,N)

END IF
*

READ (NIN,*) JOB
*
* Calculate all the eigenvalues and eigenvectors of A
*

CALL cheevd(JOB,UPLO,N,A,LDA,W,WORK,LWORK,RWORK,LRWORK,IWORK,
+ LIWORK,INFO)

*
WRITE (NOUT,*)
IF (INFO.GT.0) THEN

WRITE (NOUT,*) ’Failure to converge.’
ELSE

*
* Print eigenvalues and eigenvectors
*

WRITE (NOUT,*) ’Eigenvalues’
DO 20 I = 1, N

WRITE (NOUT,99999) I, W(I)
20 CONTINUE

WRITE (NOUT,*)
IFAIL = 0

*
CALL X04DAF(’General’,’ ’,N,N,A,LDA,’Eigenvectors’,IFAIL)

*
END IF

END IF
STOP

*
99999 FORMAT (3X,I5,5X,2F8.4)

END

9.2 Program Data

F08FQF Example Program Data
4 :Value of N
’L’ :Value of UPLO
(1.0, 0.0)
(2.0, 1.0) (2.0, 0.0)
(3.0, 1.0) (3.0, 2.0) (3.0, 0.0)
(4.0, 1.0) (4.0, 2.0) (4.0, 3.0) (4.0, 0.0) :End of matrix A
’V’ :Value of JOB

9.3 Program Results

F08FQF Example Program Results

Eigenvalues
1 -4.2443
2 -0.6886
3 1.1412
4 13.7916

Eigenvectors
1 2 3 4

1 0.4836 0.6470 -0.4456 -0.3859
0.0000 0.0000 0.0000 0.0000

2 0.2912 -0.4984 -0.0230 -0.4441
-0.3618 -0.1130 -0.5702 0.0156

3 -0.3163 0.2949 0.5331 -0.5173
-0.3696 0.3165 0.1317 -0.0844

4 -0.4447 -0.2241 -0.3510 -0.5277
0.3406 -0.2878 0.2261 -0.3168

